
Black-box definition
K number field of degree n and signature (r1, r2).

The "group of idèles of K " is a topological Abelian group A×
K

with
▶ an embedding K×

v ↪→ A×
K for every completion Kv of K ;

▶ a diagonal embedding K× ↪→ A×
K .

The quotient ("idèle class group")

CK = A×
K /K×

is isomorphic to R × a compact group.

A Hecke character is a continuous morphism

χ : CK → C×.

Finite level version

The groups CK or Hom(CK ,C×) are too big to handle
algorithmically: cut them into smaller pieces!

Modulus m: pair (mf ,m∞) = (nonzero ideal, subset of the real
embeddings).

We can define certain open subgroups U(m) of A×
K such that

▶ every Hecke character vanishes on some U(m), and
▶ Cm = A×

K /K×U(m) is of an appropriate size: a finite
dimensional manifold.

1 → R× compact torus → Cm → finite group → 1.

Finite level version

The groups CK or Hom(CK ,C×) are too big to handle
algorithmically: cut them into smaller pieces!

Modulus m: pair (mf ,m∞) = (nonzero ideal, subset of the real
embeddings).

We can define certain open subgroups U(m) of A×
K such that

▶ every Hecke character vanishes on some U(m), and
▶ Cm = A×

K /K×U(m) is of an appropriate size: a finite
dimensional manifold.

1 →
[
(R>0)

r1 × (C×)r2
]
/
[
Z×

K ∩ U(m)
]
→ Cm → Clm(K ) → 1.

Finite level version

For Hecke characters, this means:

Hom(CF ,C×) =
⋃
m

Hom(Cm,C×),

and for every m,

Hom(Cm,C×) ∼= finite × Zn−1 × C.

Finite order characters of Cm are exactly characters of Clm(K ).

Initialisation

We initialise Hom(Cm,C×) with gcharinit:

? bnf = bnfinit(polcyclo(5),1);
? gc = gcharinit(bnf,5);
? gc.cyc
% = [5, 0, 0, 0, 0.E-57]

Hom(Cm,C×) ∼= Z/5Z× Z3 × C

Conductor

The conductor of a Hecke character is the smallest m such that
χ ∈ Hom(Cm,C×).

We represent a character χ by its column vector of coordinates
corresponding to gc.cyc.

? chi = [0,0,0,5,0.1*I]~;
? gcharconductor(gc,chi)
% = [[5,4,1,4;0,1,0,0;0,0,1,0;0,0,0,1], []]
? gcharconductor(gc,4*chi)
% = [1,[]]

χ has conductor p5 and χ4 has trivial conductor.

Evaluation

Let p be a prime of K and πp a uniformiser of Kp. Using the
map K×

p → A×
K , we can evaluate χ on K×

p . Define

χ(p) = χ(πp).

This is well-defined up to χ(Z×
p ), which is a finite group. If p

does not divide the conductor of χ, it is well defined.

We evaluate Hecke characters with gchareval:

? pr11 = idealprimedec(bnf,11)[1];
? gchareval(gc,chi,pr11)
% = 0.8531383657 - 0.52168470249*I

Local characters: archimedean places

Let v be a place of K . We can restrict χ to K×
v .

Characters of R× are of the form

x 7→ sign(x)k |x |iφ

with k ∈ Z/2Z and φ ∈ C.

Characters of C× are of the form

z 7→
(

z
|z|

)k

|z|2iφ

with k ∈ Z and φ ∈ C.

Local characters: archimedean places

We obtain the local characters with gcharlocal.

Archimedean places are represented by a number between 1
and r1 + r2.

? gcharlocal(gc,chi,1)
% = [5, -0.7160628256]
? gcharlocal(gc,chi,2)
% = [0, 0.9160628256]

Local characters: nonarchimedean places

Let p be a prime of K .
A character on K×

p is completely determined by
▶ its restriction to the finite group Z×

p /(Z×
p ∩ U(m)), and

▶ its value exp(2πiθ) on πp.



Hecke Grossencharacters

Local characters: nonarchimedean places

We specify a nonarchimedean place by a prime ideal.

? pr5 = idealprimedec(bnf,5)[1];
? loc = gcharlocal(gc,chi,pr5,&bid)
% = [15, 0, 0, -0.15061499993]
? bid.cyc
% = [20, 5, 5]
? charorder(bid,loc[1..-2])
% = 4

We have Z×
p /(Z×

p ∩ U(m)) ∼= Z/20Z× (Z/5Z)2, and χ|Z×
p

has
order 4. So χ(p) is well-defined up to multiplication by a 4-th
root of unity.



Hecke Grossencharacters

L-function

Let χ be a Hecke character of conductor m. Define

L(χ, s) =
∏
p∤m

(1 − χ(p)N(p)−s)−1.

This defines an L-function:
▶ it extends to a meromorphic function on C;
▶ it satisfies a functional equation, with gamma factors given

by the (kv , φv ) at archimedean places, and of
conductor |∆K |N(m).

L-function

We can use the lfun functionalities for L-functions of Hecke
characters (currently: no imaginary component in χ).

? L = lfuncreate([gc,chi[1..-2]]);
? lfunparams(L)[1] \\conductor
% = 625
? lfunparams(L)[3]*1.
% = [5/2 - 0.8160628256*I, 0.8160628256*I,

7/2 - 0.8160628256*I, 1 + 0.8160628256*I]
? lfuncheckfeq(L)
% = -132
? lfun(L,1)
% = 1.0185518145 + 0.1382746268*I

Algebraic characters

A Hecke character is called algebraic if for every complex
embedding σ, there exists pσ,qσ such that for all z ∈ (K×

σ )◦,

χ(z) = z−pσ(z̄)−qσ .

We then say that χ is of type ((pσ,qσ))σ.

Equivalently, there exists a number field E such that for all p,

χ(p) ∈ E×.

Algebraic characters
We can test the algebraicity of a character and compute its type
with gcharisalgebraic:

? gcharisalgebraic(gc,chi)
% = 0
? chi2 = [0,1,0,0,0]~
? gcharisalgebraic(gc,chi2,&typ)
% = 1
? typ
% = [[-1, 1], [0, 0]]
? gcharlocal(gc,chi2,1)
% = [2, 0]
? gcharlocal(gc,chi2,2)
% = [0, 0]

χ is not algebraic, but χ2 is algebraic of type ((−1,1), (0,0)).



Hecke Grossencharacters

Algebraic characters

The set of algebraic characters of modulus m is a finitely
generated group.
We can compute a basis of this group with gcharalgebraic:

? gcharalgebraic(gc)
% = [1 0 0 0]

[0 1 0 0]
[0 0 1 0]
[0 0 0 0]
[0 0 -1/2 -1]

Algebraic characters

Every finite order Hecke character is algebraic, and the type of
an algebraic character determines it up to multiplication by a
finite order character.
We can search for an algebraic character of a given type with
gcharalgebraic(gc,type):

? gcharalgebraic(gc,[[1,2],[3,4]])
% = []
? gcharalgebraic(gc,[[2,-2],[-1,1]])
% = [[0, -1, 2, 0, 0]~]

There is no character of type ((1,2), (3,4)), but we found a
character of type ((2,−2), (−1,1)).

Identification

We can look for a character given some information about its
values or its local characters with gcharidentify.

? pr31 = idealprimedec(bnf,31)[1];
? gcharidentify(gc,[pr11,pr31],[0.261946,-0.497068])
% = [3, -77916, 53772, 206992]~

This is probably meaningless because the number of digits of
the output is of the same order as the precision we had on the
values.

Identification

We need to reduce the working precision:

? localprec(6); chi3=gcharidentify(gc,[pr11,pr31],
[0.261946,-0.497068])

% = [0, -3, 2, 8]~
? gchareval(gc,chi3,pr11,0)
% = 0.26194591587002798940182987097135921818
? gchareval(gc,chi3,pr31,0)
% = -0.49706763230668562700776309783089085752

Identification

To ensure reliable identification, even with low precision, you
need to provide all archimedean places and the values at a set
of primes that generates the ray class group Clm(K ).

? chi4 = gcharidentify(gc,[1,2,pr11],[[-26,-0.1],
[13,0.1],0.])

% = [1, -7, 13, 1]~
? gcharlocal(gc,chi4,1)
% = [-26, -0.1632125651]
? gcharlocal(gc,chi4,2)
% = [13, 0.1632125651]
? gchareval(gc,chi4,pr11)
% = 0.9007070934 - 0.4344269003*I

Example: CM abelian surface
By CM theory, the L-function of every CM abelian varietie is a
product of L-functions of algebraic Hecke characters.
Let’s compute an example: consider the genus 2 curve

C : y2 + x3y = −2x4 − 2x3 + 2x2 + 3x − 2

and let A be its Jacobian.

? C = [-2*x^4 - 2*x^3 + 2*x^2 + 3*x - 2, x^3];
? L = lfungenus2(C);
? lfunparams(L)
% = [28561, 2, [0, 0, 1, 1]]
? factor(lfunparams(L)[1])
% = [13 4]

A has good reduction outside 13.

Example: CM abelian surface

E = bnfinit(y^4 - y^3 + 2*y^2 + 4*y + 3, 1);
poldegree(nfsubfieldscm(E)[1])
% = 4

The maximal CM subfield of E has degree 4, i.e. E is a CM
field. It is known that A has CM by E .
We would like an associated Hecke character.

? pr13 = idealprimedec(E,13)[1];
? gc2 = gcharinit(E,pr13);
? gc2.cyc
% = [3, 0, 0, 0, 0.E-57]
? chiC = [1, -1, -1, 0, -1/2]~

Example: CM abelian surface

? gcharisalgebraic(gc2,chiC,&typ)
? typ
% = [[1, 0], [1, 0]]

This is the type we expect for an algebraic Hecke character
corresponding to an abelian variety.

? L2 = lfuncreate([gc2,chiC]);
? lfunparams(L2)
% = [28561, 2, [0, 0, 1, 1]]
? exponent(lfunan(L,1000)-lfunan(L2,1000))
% = -120

The L-functions match!

Example: density

For varying conductor, the possible parameters at infinity of
Hecke characters are dense.

? gc3 = gcharinit(x^3-3*x+1,2^20);
? chiapprox = gcharidentify(gc3,[1,2,3],[[0,Pi],
[0,exp(1)],[0,-Pi-exp(1)]])

% = [0, 1338253, 2033118]~
? gcharlocal(gc3,chiapprox,1)
% = [0, 3.141592238]
? gcharlocal(gc3,chiapprox,2)
% = [0, 2.718283147]

For this χ, we have φ1 ≈ π and φ2 ≈ e!

Example: partially algebraic characters
The algebraicity of a Hecke character is almost equivalent to
the vanishing of all φσ parameters.
There also exists character for which a subset of the φσ vanish.

? gc4 = gcharinit(x^4-5,1);
? gc4.cyc
% = [0, 0, 0, 0.E-57]
? chipart = [1,0,0,0]~;
? gcharlocal(gc4,chipart,1)
% = [0, 0.7290851962]
? gcharlocal(gc4,chipart,2)
% = [0, -0.7290851962]
? gcharlocal(gc4,chipart,3)
% = [-2, 0.E-95]

For this χ, we have φ1, φ2 ̸= 0 but φ3 = 0!

Questions ?

Have fun with GP !

Implementation based on
https://inria.hal.science/hal-03795267.

https://inria.hal.science/hal-03795267

